QT +=core gui quick virtualkeyboard qml quickcontrols2 charts

TARGET = App
TEMPLATE = app

CONFIG += qml_debug
CONFIG += debug


# You can make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0
include(KeyBoard/KeyBoard.pri)
include(TaoQuick/TaoQuick.pri)
include(TaoCommon/TaoCommon.pri)
DEFINES += QMAKE_GEN_TAOMACRO
!build_pass {
    headerContents = \
        "$${LITERAL_HASH}pragma once" \
        "$${LITERAL_HASH}define TaoQuickImport $${TaoQuickImport}" \
        "$${LITERAL_HASH}define TaoQuickImage $${TaoQuickImage}" \
        "$${LITERAL_HASH}define qmlPath $${qmlPath}" \
        "$${LITERAL_HASH}define contentsPath $${contentsPath}" \
        "$${LITERAL_HASH}define imgPath $${imgPath}" \
        "$${LITERAL_HASH}define transDir $${transDir}"

    write_file(taoMacro.h, headerContents)
}

SOURCES += \
        main.cpp \
        s401model.cpp \
        sensorlist.cpp \
        sensormodel.cpp

RESOURCES += qml.qrc



disable-desktop|android-embedded|!isEmpty(CROSS_COMPILE)|qnx {
    DEFINES += MAIN_QML=\\\"basic-b2qt.qml\\\" # 采用了自定义的虚拟键盘的InputPanel，可以依附窗口。
} else {
    DEFINES += MAIN_QML=\\\"Basic.qml\\\"      # 独立于窗口
}

DEFINES += imgPATH=\\\"qrc:/SutoDisplay/content/images/\\\"      # 独立于窗口
DEFINES += conPATH=\\\"qrc:/SutoDisplay/content/usercon/\\\"      # 独立于窗口
# Additional import path used to resolve QML modules in Qt Creator's code model
QML_IMPORT_PATH +=$$PWD/SutoDisplay/imports

# Additional import path used to resolve QML modules just for Qt Quick Designer
QML_DESIGNER_IMPORT_PATH +=$$PWD
QML_DESIGNER_IMPORT_PATH +=$$PWD/SutoDisplay/imports
QML_DESIGNER_IMPORT_PATH +=$$PWD/SutoDisplay/content

target.path += /root
INSTALLS += target

DISTFILES +=

HEADERS += \
    def.h \
    s401model.h \
    sensorlist.h \
    sensormodel.h
